
Subject: Re: object graphics question

Posted by [btt](#) on Wed, 04 Feb 2004 14:54:03 GMT

[View Forum Message](#) <> [Reply to Message](#)

Karsten Rodenacker wrote:

```
>
> I have a problem with coordinated transformations of IDLgrImage and
> other objects, e.g. IDLgrContour.
>
> Does anybody know why rotating behaves like it does?
> In the small example:
>
> img=read_png( filepath('mineral.png', SUBDIR=['examples','data']))
> m=obj_new('IDLgrModel')
> m->add,obj_new('IDLgrImage',img)
> m->add,obj_new('IDLgrContour',img,color=[255,0,0])
> xobjview,m,/block
>
> rotation shows a quite bizar behaviour.
>
> I would expect that the image moves like one level of the contour object.
>
```

Hi Karsten,

Yes, that is wierd... but there is a solution.

You should really be rendering a polygon with the image as a texture map. IDLgrImage are always displayed at Z=0 (or some such place.) To give the image a real coordinate system you have to attach it to some spatial object, like a polygon or a surface.

```
img=read_png( filepath('mineral.png', SUBDIR=['examples','data']))
m=obj_new('IDLgrModel')
oimg = obj_new('IDLgrImage',img)
oimg -> GetProperty, dim = dim
poly = obj_new("IDLgrPolygon", [0,0,dim[0]-1, dim[0]-1], $
[0,dim[1]-1,dim[1]-1, 0], $
color = [255,255,255], $
Texture_Map = oimg,$
Texture_Coord = [[0,0],[0,1], [1,1], [1,0]])
m->add, poly
m->add,obj_new('IDLgrContour',img,color=[255,0,0])
xobjview,m,/block
```

Cheers,

Ben
